


5.5 Überlagerte Konturen

Die Steuerung betrachtet eine programmierte Kontur als Tasche. Mit den Funktionen der Konturformel haben Sie die Möglichkeit, eine Kontur in eine Insel umzuwandeln.

Taschen und Inseln können Sie zu einer neuen Kontur überlagern. Damit können Sie die Fläche einer Tasche durch eine überlagerte Tasche vergrößern oder eine Insel verkleinern.



i Die TNC 640 interpretiert die überlagerten Flächen anders als die iTNC 530. Bei der TNC 640 hat eine eingegebene Tiefe bei einer Insel keine Wirkung.

Beispiel: Überlagerte Taschen

i Die nachfolgenden Beispiele sind Konturbeschreibungsprogramme, die in einem Konturdefinitionsprogramm definiert sind. Das Konturdefinitionsprogramm wiederum ist über die Funktion **SEL CONTOUR** im eigentlichen Hauptprogramm aufzurufen.

- Die Taschen A und B überlagern sich.
- Die Steuerung berechnet die Schnittpunkte S1 und S2 selbst.
- Die Taschen sind als Vollkreise programmiert.
- Die Flächen A und B müssen in separaten NC-Programmen ohne Radiuskorrektur programmiert sein

Konturbeschreibungsprogramm 1: Tasche A

```
0 BEGIN PGM TASCHE_A MM
1 L X+10 Y+50 R0
2 CC X+35 Y+50
3 C X+10 Y+50 DR-
4 END PGM TASCHE_A MM
```

Konturbeschreibungsprogramm 2: Tasche B

```
0 BEGIN PGM TASCHE_B MM
1 L X+90 Y+50 R0
2 CC X+65 Y+50
3 C X+90 Y+50 DR-
4 END PGM TASCHE_B MM
```

Summenfläche

Beide Teilflächen A und B inklusive der gemeinsam überdeckten Fläche sollen bearbeitet werden:

- In der Konturformel werden die Flächen A und B mit der Funktion "vereinigt mit" verrechnet


51 ...

52 DECLARE CONTOUR QC1 = "TASCHE_A.H"

53 DECLARE CONTOUR QC2 = "TASCHE_B.H"

54 QC10 = QC1 | QC2

55 ...



Differenzfläche

Fläche A soll ohne den von B überdeckten Anteil bearbeitet werden:

- In der Konturformel wird die Fläche B mit der Funktion **ohne** von der Fläche A abgezogen


51 ...

52 DECLARE CONTOUR QC1 = "TASCHE_A.H"

53 DECLARE CONTOUR QC2 = "TASCHE_B.H"

54 QC10 = QC1 \ QC2

55 ...



Schnittfläche

Die von A und B überdeckte Fläche soll bearbeitet werden:

- In der Konturformel werden die Flächen A und B mit der Funktion "geschnitten mit" verrechnet


51 ...

52 DECLARE CONTOUR QC1 = "TASCHE_A.H"

53 DECLARE CONTOUR QC2 = "TASCHE_B.H"

54 QC10 = QC1 & QC2

55 ...



5.6 Aufgabe Scheibe Zeichnungsnummer 1265444

Text:		ID number	
		Change No.	C000941-05
		Phase:	Nicht-Serie
	Original drawing	Scheibe Washer	
RoHS	Scale Format		
1:1	A4	Werkstoff: Material:	
Maße in mm / Dimensions in mm		Einzelteilzeichnung / Component Drawing	
Werkstückkanten nach ISO 13715 Workpiece edges ISO 13715 	Allgmeintoleranzen ISO 2768-mH $\leq 6\text{mm}: \pm 0,2$ General tolerances ISO 2768-mH $\leq 6\text{mm}: \pm 0,2$	Tolerierung nach ISO 8015 Tolerances as per ISO 8015	●blanke Flächen/Blank surfaces Oberflächen nach ISO 1302 Surfaces as per ISO 1302
		Oberflächenbehandlung: Surface treatment:	
The reproduction, distribution and utilization of this document as well as the communication of its contents to others without express authorization is prohibited. Offenders will be held liable for the payment of damages. All rights reserved in the event of the grant of a patent, utility model or design. (ISO 16016)			
HEIDENHAIN DR. JOHANNES HEIDENHAIN GmbH 83301 Traunreut, Germany	Created	Responsible	Released
	Feichtner		
	31.08.2018		
		Version	Revision Sheet Page
		D1265444-00-A-01	
		Document number	
		1 of 1	

Komplexe Konturen zusammensetzen

Um eine Konturformel zu erstellen, ist es hilfreich, die unterschiedlichen Flächen mit ihren Tiefen zu betrachten.

Tiefste Tasche I

```
DECLARE CONTOUR QC1 = "Circle_D60.H" DEPTH9
```

Mittlere Tasche II

```
DECLARE CONTOUR QC2 = "Circle_D40.H" DEPTH6
```

Kleine Tasche III


```
DECLARE CONTOUR QC3 = "Circle_D20.H" DEPTH3
```

Rechte Tasche IV

```
DECLARE CONTOUR QC4 = "Circle_D20_RIGHT.H" DEPTH1
```

Linke Tasche V

```
DECLARE CONTOUR QC5 = "Circle_D20_LEFT.H" DEPTH1
```



Kontur zusammenfügen

Formel für die Tiefe -9

Die tiefste Tasche QC1 soll auf ihrer Tiefe ohne die Insel QC2 ausgeräumt werden.

```
QC10 = QC1 \ QC2
```

Formel für die Tiefe -6

Die mittlere Tasche soll ohne die Insel QC3 ausgeräumt werden.

```
QC10 = QC2 \ QC3
```

Formel für die Tiefe -3

Die kleine Tasche soll ohne die Schnittmenge der linken und rechten Insel QC4 und QC5 ausgeräumt werden.

```
QC6 = QC4 & QC5
```

```
QC10 = QC3 \ QC6
```

Formel für die Tiefe -1

Die kleinste Tasche soll komplett ausgeräumt werden.


```
QC10 = QC6
```

Zusammengesetzt ergibt das die Formel für die komplette Kontur.

```
QC10 = ( QC1\QC2 ) | ( QC2 \ QC3 ) | ( QC3 \ QC6 ) | QC6
```

Arbeitsplan

- ▶ Konturprogramme der einzelnen Konturen erstellen, z. B.
 - Circle_D20.h
 - Circle_D40.h
 - Circle_D60.h
 - Circle_D20_RIGHT.h
 - Circle_D20_LEFT.h
- ▶ NC-Programm zum Verknüpfen der Konturen erstellen
- ▶ Hauptprogramm erstellen
- ▶ Mit **SEL CONTOUR** Konturbeschreibung wählen
- ▶ Ausräumen mit SL-Zyklen
 - Werkzeugaufruf, z. B. MILL_D8_ROUGH
 - Zyklus 14 **KONTUR**
 - Zyklus 270 **KONTURZUG-DATEN**
 - Zyklus 25 **KONTUR-ZUG** mit **AUFMASS SEITE**
 - Zyklusaufruf
- ▶ NC-Programm beenden



Lösung 1265444

0 BEGIN PGM 1265444 MM	
1 BLK FORM CYLINDER Z D80 L20	
2 ;	
3 * -	Ausraeumen
4 TOOL CALL "MILL_D8_ROUGH" Z S18000	
5 L Z+100 R0 FMAX M3	
6 SEL CONTOUR "CONTOUR_1265444.h"	
7 CYCL DEF 20 KONTUR-DATEN	
Q1=-20 ;FRAESTIEFE	
Q2=+1 ;BAHN-UEBERLAPPUNG	
Q3=+0 ;AUFMASS SEITE	
Q4=+0 ;AUFMASS TIEFE	
Q5=+0 ;KOOR. OBERFLAECHE	
Q6=+2 ;SICHERHEITS-ABST.	
Q7=+50 ;SICHERE HOEHE	
Q8=+1 ;RUNDUNGSRADIUS	
Q9=+1 ;DREHSINN	
8 CYCL DEF 22 AUSRAEUMEN	
Q10=-5 ;ZUSTELL-TIEFE	
Q11=+150 ;VORSCHUB TIEFENZ.	
Q12=+2500 ;VORSCHUB RAEUMEN	
Q18=+0 ;VORRAEUM-WERKZEUG	
Q19=+2500 ;VORSCHUB PENDELN	
Q208=+99999 ;VORSCHUB RUECKZUG	
Q401=+75 ;VORSCHUBFAKTOR	
Q404=+0 ;NACHRAEUMSTRATEGIE	
9 CYCL CALL	
10 M30	
11 END PGM 1265444 MM	

Konturprogramme 1265444

0 BEGIN PGM CIRCLE_D20 MM	
1 CC X+0 Y+0	
2 LP PR+10 PA+0	
3 CP IPA+360 DR+	
4 END PGM CIRCLE_D20 MM	
0 BEGIN PGM CIRCLE_D40 MM	
1 CC X+0 Y+0	
2 LP PR+20 PA+0	
3 CP IPA+360 DR+	
4 END PGM CIRCLE_D40 MM	
0 BEGIN PGM CIRCLE_D60 MM	
1 CC X+0 Y+0	
2 LP PR+30 PA+0	
3 CP IPA+360 DR+	
4 END PGM CIRCLE_D60 MM	
0 BEGIN PGM CIRCLE_D20_RIGHT MM	
1 CC X+5 Y+0	
2 LP PR+10 PA+0	
3 CP IPA+360 DR+	
4 END PGM CIRCLE_D20_RIGHT MM	
0 BEGIN PGM CIRCLE_D20_LEFT MM	
1 CC X-5 Y+0	
2 LP PR+10 PA+0	
3 CP IPA+360 DR+	
4 END PGM CIRCLE_D20_LEFT MM	
0 BEGIN PGM CONTOUR_1265444 MM	
1 * -	Konturprogramme mit Tiefe festlegen
2 DECLARE CONTOUR QC1 = "Circle_D60.H" DEPTH9	
3 DECLARE CONTOUR QC2 = "Circle_D40.H" DEPTH6	
4 DECLARE CONTOUR QC3 = "Circle_D20.H" DEPTH3	
5 DECLARE CONTOUR QC4 = "Circle_D20_RIGHT.H" DEPTH1	
6 DECLARE CONTOUR QC5 = "Circle_D20_LEFT.H" DEPTH1	
7 * -	Konturprogramme mit Formel zusammenfassen
8 QC6 = QC4 & QC5	
9 QC4 = (QC1 \ QC2) (QC2 \ QC3) (QC3 \ QC6) QC6	
10 END PGM CONTOUR_1265444 MM	